

ОПИС РУХУ N-ЛАНКОВОГО МАЯТНИКА ЗА ДОПОМОГОЮ ОПЕРАТОРІВ СИСТЕМИ MAPLE

Університет цивільного захисту України (м. Харків)
Київський національний університет будівництва і архітектури

Розроблено спосіб складання у середовищі математичної системи комп'ютерної алгебри Maple диференціальних рівнянь опису руху n-ланкового маятника. Як узагальнені змінні, залежні від часу, обрано кути, утворені ланками маятника з вертикалями.

Постановка проблеми. У 1788 році Лагранж застосував варіаційний принцип до визначення загальної кінетичної й потенціальної енергії механічної системи з урахуванням її кінематичних зв'язків. У результаті Лагранж одержав універсальний підхід для запису рівнянь руху будь-якої механічної системи у вигляді рівнянь руху, відомих як рівняння Лагранжа II роду. Але трудомісткість виводу рівнянь Лагранжа виявилася дуже високою, а їхній запис у явній формі занадто громіздкий. Тому за відсутності обчислювальної техніки реальне застосування рівнянь Лагранжа обмежується відносно простими випадками [1].

Пояснимо це на прикладі n-ланкового маятника. Вважатимемо, що цей маятник складається з невагомих нерозтяжних стержнів довжин L_i ($i=1..n$), шарнірно сполучених між собою прикінцевими вузловими точками, на яких закріплено кульки з масами m_i ($i=1..n$). Узагальненими координатами є кути u_i ($i=1..n$), утворені відповідними ланками з вертикалями (рис. 1). Для спрощення вважають, що тертя руху відсутнє, а точка кріплення нерухома.

Опис коливання маятника за умови відсутності дисипативних сил виконується на основі рівнянь Лагранжа II роду

$$\frac{\partial}{\partial t} \left(\frac{\partial}{\partial u_i'} L \right) - \frac{\partial}{\partial u_i} (L) = 0, \quad (i=1..n), \quad (1)$$

де $L = K(n) - P(n)$ - лагранжіан; $K(n)$ - кінетична енергія системи; $P(n)$ - потенціальна енергія системи; u_i - i-та узагальнена координата (кут між вертикаллю і ланкою); $u_i' = \frac{d}{dt} u_i$.

Для обчислення кінетичної та потенціальної енергії маємо вирази [2, 3]:




Рис. 1. Схема n-ланкового маятника

$$K(n) = \frac{1}{2} \sum_{k=1}^n m_k \left[\left(-\sum_{i=1}^{k-1} L_i \cos(u_i(t)) \frac{du_i(t)}{dt} - L_k \cos(u_k(t)) \frac{du_k(t)}{dt} \right)^2 + \left(\sum_{i=1}^{k-1} -L_i \sin(u_i(t)) \frac{du_i(t)}{dt} - L_k \sin(u_k(t)) \frac{du_k(t)}{dt} \right)^2 \right]; \quad (2)$$

$$P(n) = g \sum_{k=1}^n m_k \left(\sum_{i=1}^{k-1} L_i \cos(u_i(t)) + L_k \cos(u_k(t)) \right). \quad (3)$$

В результаті опис руху n -ланкового маятника одержують у вигляді системи з n диференціальних рівнянь відносно кутів $u_i(t)$ ($i=1..n$), складених за допомогою рівнянь Лагранжа II роду (1).

Метод рівнянь Лагранжа II роду визначає строгий підхід, який полягає у формальному записі виразів для кінетичної і потенціальної енергії з наступним їх диференціюванням. При цьому складність програмної реалізації рівнянь Лагранжа II роду викликана особливістю виразу (1), де необхідно визначити два види похідних: першого - за часом t , а другого - «за похідною u_i' ». Тому за умови програмної реалізації рівнянь Лагранжа II роду конкретною алгоритмічною мовою слід намагатися засобами її операторів автоматизувати процес складання зазначених рівнянь. У роботі дану проблему розв'язано на прикладі мови Maple.

Огляд літературних джерел. Для складання та розв'язання у замкнутому вигляді систем диференціальних рівнянь вигляду (1) необхідні оператори конкретної мови програмування та обчислювальні ресурси комп'ютерів. Були створені інтерактивні інструменти для моделювання, імітації й аналізу динамічних систем, наприклад, MathLab/Simulink, Adams/modeFrontier, «Універсальний механізм» тощо [4]. Вони мають певні недоліки при розробці моделей динамічних систем із великою кількістю диференціальних рівнянь, особливо коли важлива швидкість їхнього виконання. Крім того часто існують обмеження на величини кутів відхилень ланок маятників. Тому доцільним буде визначення коливань n -ланкового маятника, яке базується на безпосередньому розв'язанні системи диференціальних рівнянь відносно узагальнених координат маятника - кутів, утворених відповідними ланками з вертикалями.

У роботах [5, 6] наведено метод автоматизованого формування системи двох (як приклад) диференціальних рівнянь Лагранжа II роду

$$\frac{d}{dt} \left(\frac{\partial L}{\partial u'} \right) - \frac{\partial L}{\partial u} = 0; \quad \frac{d}{dt} \left(\frac{\partial L}{\partial v'} \right) - \frac{\partial L}{\partial v} = 0, \quad (4)$$

де u' і v' - похідні за часом t від узагальнених координат u і v , $L = K - P$, де позначено через K кінетичну енергію, а через P - потенціальну енергію, яка може задаватися узагальненими силами.

Для автоматизованого формування у аналітичному вигляді системи рівнянь Лагранжа другого роду було використано оператор **subs (X=A, B)**, який у виразі B заміняє підвираз X на підвираз A . Тоді результати

диференціювання $\frac{\partial L}{\partial u'}$, $\frac{d}{dt}\left(\frac{\partial L}{\partial u'}\right)$ і $\frac{\partial L}{\partial u}$ в аналітичному вигляді було одержано

за допомогою Maple-операторів

```
dL_U := subs(W = diff(u(t), t),  
diff(subs(diff(u(t), t) = W, L), W));
```

```
dL_U_dt := diff(dT_U, t);
```

```
dL_U_ := subs(W = u(t), diff(subs(u(t) = W, L), W)).
```

У результаті чого рівняння Лагранжа II роду в аналітичному вигляді було одержано так: $dL_U_dt - dL_U_ = 0$.

З урахуванням координат вузлових точок (подвійний маятник, $n=2$)

```
xa := L1*sin(u(t));  
ya := L1*cos(u(t));  
xb := xa+L2*sin(v(t));  
yb := ya+L2*cos(v(t)),
```

а також з урахуванням виразів для кінетичної та потенціальної енергії, рівняння Лагранжа II роду за синтаксисом Maple одержано у вигляді:

$$L := \frac{1}{2} m1 \left(L1^2 \cos(u(t))^2 \left(\frac{d}{dt} u(t) \right)^2 + L1^2 \sin(u(t))^2 \left(\frac{d}{dt} u(t) \right)^2 \right) + \frac{1}{2} m2 \left(\left(L1 \cos(u(t)) \left(\frac{d}{dt} u(t) \right) + L2 \cos(v(t)) \left(\frac{d}{dt} v(t) \right) \right)^2 + \left(-L1 \sin(u(t)) \left(\frac{d}{dt} u(t) \right) - L2 \sin(v(t)) \left(\frac{d}{dt} v(t) \right) \right)^2 \right) + (m1 + m2) g L1 \cos(u(t)) + m2 g (L1 \cos(u(t)) + L2 \cos(v(t))). \quad (5)$$

У роботах [5, 6] показано, як вираз (5) звести до системи двох рівнянь, як її чисельно розв'язати з початковими умовами та як одержані розв'язки можна трактувати і використовувати, як аналітичні вирази.

Недолік результатів робіт [5, 6] полягає у тому, що рівняння Лагранжа II роду можна програмно реалізувати лише для окремих значень n . У разі інших значень, програмну реалізацію слід проводити заново.

Постановка завдання. Розробити універсальний спосіб програмної реалізації рівнянь Лагранжа II роду для довільних значень n , коли узагальненими координатами маятника є невідомі кути $u_1(t)$, $u_2(t)$, ..., $u_n(t)$, утворені відповідними ланками з вертикалями.

Основна частина. Для виконання Maple-програми необхідно підготувати початкові дані:

- n -кількість маятників;
 - вектор довжин ланок маятника: $\mathbf{L} = \{L_1, L_2, \dots, L_n\}$;
 - вектор значень мас кульок: $\mathbf{m} = \{m_1, m_2, \dots, m_n\}$;
 - вектор початкових кутів відхилень: $\bar{\boldsymbol{\theta}} = \{ \bar{u}_1(0), \bar{u}_2(0), \dots, \bar{u}_n(0) \}$.
- Необхідно знайти вектор кутів відхилень: $\boldsymbol{\theta} = \{u_1(t), u_2(t), \dots, u_n(t)\}$.

Визначаємо компоненти радіус-векторів і швидкостей для кожної маси у вигляді списку:

```

x := [ L[1]*sin(u[1](t)) ];
y := [-L[1]*cos(u[1](t)) ];
for i from 2 to n do
x := [op(x), x[i-1] + L[i]*sin(u[i](t))];
y := [op(y), y[i-1] - L[i]*cos(u[i](t))];
od:

```

Оператор **op** призначено для «витагування» операндів із виразів, а оператором **seq** створюються послідовності виразів.

```

r := [seq([x[i],y[i]],i=1..n)]:
vv := map(x -> map(diff,x,t), r):

```

Оператор **map** застосовує процедуру до кожного операнда зі списку. Наприклад, **map(x -> x^2, x + y)** означає $x^2 + y^2$.

Далі необхідно створити послідовності для формування виразів у загальному вигляді, залежних від n :

```

seq_P := seq(u[i](t), i=1..n):
seq_V := seq(diff(u[i](t), t), i=1..n):
seq_A := seq(diff(u[i](t), t, t), i=1..n):
seq_M := seq(m[i], i=1..n):
seq_L := seq(L[i], i=1..n):

```

Для обчислення кінетичної енергії визначимо функцію, яку буде відображено на швидкостях переміщення вузлових точок:

```

v1 := x -> combine(collect(x[1]^2 + x[2]^2,
[seq(diff(u[i](t), t), i=1..n), seq(L[i], i=1..n)])):

```

Тут оператор **combine** об'єднує декілька виразів у один, а оператор **collect** збирає коефіцієнти при однакових степенях. У результаті одержуємо вираз для визначення квадрата швидкості:

```

v2 := map(v1, vv);

```

Далі сформуємо вирази у канонічному вигляді для кінетичної і потенціальної енергії, де оператором **add** підсумовується вираз:

```

K := add(1/2*m[i]*v2[i], i=1..n);
P := add((add(L[j], j=1..i) + y[i])*m[i]*g, i=1..n);

```

У результаті одержуємо вираз опису рівняння Лагранжа II роду:

```

Lag := K - P;

```

Перетворити з канонічного до практичного вигляду вирази кінетичної і потенціальної енергії необхідно за допомогою підстановок:

```

Lag1 := [seq(diff(subs(qd=diff(u[i](t), t),
diff(subs(diff(u[i](t), t)=qd, Lag), qd)), t), i=1..n)]:
Lag1 := [seq(Lag1[i] - subs(qit=u[i](t),
diff(subs(u[i](t)=qit, Lag), qit)), i=1..n)]:
LAG := map(collect, Lag1,
[seq_A, seq_V, seq_M, seq_L]):

```

Таким чином було сформовано вираз для складання системи диференціальних рівнянь Лагранжа II роду для довільного значення n .

Для розв'язання системи диференціальних рівнянь задамо початкові умови у вигляді:

```
seq_CIP := seq(u[i](0)=u0[i], i=1..N);
seqCIV := seq(D(u[i])(0)=0, i=1..N);
```

Перед розв'язанням системи рівнянь виконаємо її перетворення так, щоб у лівій частині кожного рівняння залишилися лише другі похідні. Для цього використаємо оператор **genmatrix** із бібліотеки **linalg**:

```
seq_LS := seq(seq_A[i] = z[i], i=1..N);
A := genmatrix(subs(seq_LS, LAG),
               [seq(z[i], i=1..N)], b);
zz := linsolve(A, b):
```

Наближений розв'язок $\{u_1(t), u_2(t), \dots, u_n(t)\}$ системи диференці-альних рівнянь одержимо за допомогою оператора **dsolve**:

```
EDSol := dsolve({seq(seq_A[i]=zz[i], i=1..N),
                 seq_CIP, seq_CIV}, {seq_P},
               numeric, method=rkf45, output=listprocedure);
```

Тоді розв'язок буде сформовано у вигляді рядка з $2n+1$ компонентами:

$$[t, u_1(t), \frac{d}{dt}u_1(t), u_2(t), \frac{d}{dt}u_2(t), \dots, u_n(t), \frac{d}{dt}u_n(t)].$$

При цьому слід зауважити, що в процесі розв'язанні системи рівнянь ми довіряємо розробникам програмного забезпечення Maple, не вникаючи у сутність методів їх математичного наповнення й алгоритмів розв'язання даної задачі. Ця довіра базується на позитивних відгуках великої кількості користувачів Maple після багаторічної роботи із системою, а також, зважаючи на те, що програмні модулі Maple закладено в інші комп'ютерні програми, де система комп'ютерної алгебри Maple пройшла всебічну перевірку.




Рис. 2. Схема єднання строп і вантажу

Приклад (для $n = 5$). Розглянемо

реальний випадок застосування п'ятиланкового маятника на прикладі схеми з'єднання строп і вантажу (рис. 2). Нехай після аварії точка F від'єдналася від точки A і п'ятиланкова система відрізків почала рухатися в полі тяжіння. Тобто складеться ситуація, подібна коливанию п'ятиланкового маятника з прикінцевою точкою F і з вектором початкових кутів $\theta \equiv \{0, -\alpha, -\alpha, \pi/2, \pi + \alpha, \pi + \alpha\}$, де

$$\alpha = \arcsin\left(\frac{L_3}{2(L_1 + L_2)}\right).$$

Нехай $\mathbf{L} = \{1, 1, 3, 1, 1\}$ і $\mathbf{m} = \{1, 10, 10, 1, 1\}$. На рис. 3 зображено сліди переміщення ланок маятника за певні проміжки обчислювальних кроків.




Рис. 3. Сліди переміщення ланок маятника за проміжки кроків:
 а - $t = 0.25$; б - $t = 0.50$; в - $t = 0.75$; г - $t = 0.100$.

Висновок. Потреба в алгоритмах виводу рівнянь руху механічних систем спонукала до розвитку нових технологій програмування. У роботі показано, як сучасні вимоги до розрахунку механічних систем можна реалізувати засобами алгоритмічної мови Maple, цим самим забезпечивши можливість аналізу результатів і візуалізації об'єкта дослідження.

Література

1. Salisbury K.L., Knight D.G. The multiple pendulum problem via Maple.

Intern. Journal of Math. Educ. in Science and Technol., Vol. 33, Iss. 5, 2002. -

2. *Gmiterko A., Grossman M.* N-link Inverted Pendulum Modeling. Recent Advances in Mechatronics 2010, Part 3, p. 151-156.

3. *Gmiterko A, Grossman M.* An n-link inverted pendulum modeling. Acta Mechanica Slovaca, #13(3), 2009, p. 22-29

4. *Сотников А.Л.* Компьютерно-ориентированный автоматический синтез уравнений движения механических систем. / А.Л. Сотников // *Металлургические процессы и оборудование.* № 3(9), 2007, стр. 34-51

5. *Адашевська І.Ю.* Геометричне моделювання коливань багатоланкових маятникових механічних систем. Автореф. дис... канд. техн. наук: 05.01.01/ *І.Ю.Адашевська* - Київськ. нац. унів. будівн. і архіт.– Київ,2006.–20 с.

6. *Куценко Л.М.* Геометричне моделювання коливань багатоланкових маятників. / *Л.М.Куценко, І.Ю.Адашевська*– Харків:«НТМТ», 2008.– 176 с.

ОПИСАНИЕ ДВИЖЕНИЯ N-ЗВЕННОГО МАЯТНИКА ПРИ ПОМОЩИ ОПЕРАТОРОВ СИСТЕМЫ MAPLE

Л. Н. Куценко, Р. М. Колочавин

Разработан способ составления в среде математической системы компьютерной алгебры Maple дифференциальных уравнений описания движения n -звенного маятника. В качестве обобщенных переменных, зависящих от времени, выбраны углы, образованные звеньями маятника с вертикалями.

DESCRIPTION OF MOVEMENT OF A N-LINK PENDULUM BY MEANS OF OPERATORS OF MAPLE SYSTEM

L. N. Kutsenko, R. M. Kolochavin

The compilation method in the environment of mathematical system of computer algebra of Maple of differential equations of the description of movement of n – link pendulum is developed. As the generalized variables dependent on time, the angles formed by links of a pendulum with verticals are selected.